<commit_before>function prompt_char {
	if [ $UID -eq 0 ]; then echo "#"; else echo $; fi
}

PROMPT='%{$fg[cyan]%}%n %{$fg[green]%}%~ $(git_prompt_info)%{$reset_color%}$(prompt_char) '

ZSH_THEME_GIT_PROMPT_PREFIX="%{$fg[cyan]%}"
ZSH_THEME_GIT_PROMPT_SUFFIX="%{$fg[green]%} "
<commit_msg>Rewrite lime.plugin.zsh without oh-my-zsh<commit_after>prompt_lime_git() {
  # Get VCS information
  vcs_info

  # Store working_tree without the 'x' prefix
  local working_tree="${vcs_info_msg_1_#x}"
  [[ -n $working_tree ]] || return

  echo "%F{109}${vcs_info_msg_0_}$(prompt_lime_git_dirty)%f "
}

prompt_lime_git_dirty() {
  test -z "$(command git status --porcelain --ignore-submodules=dirty -unormal)"
  (( $? )) && echo "*"
}

prompt_lime_symbol() {
  if [ $UID -eq 0 ]; then
    echo '#'
  else
    echo '$'
  fi
}

prompt_lime_setup() {
  autoload -Uz vcs_info

  zstyle ':vcs_info:*' enable git
  # Export only two msg variables from vcs_info
  zstyle ':vcs_info:*' max-exports 2
  # %s: The current version control system, like 'git' or 'svn'
  # %r: The name of the root directory of the repository
  # #S: The current path relative to the repository root directory
  # %b: Branch information, like 'master'
  # %m: In case of Git, show information about stashes
  # %u: Show unstaged changes in the repository (works with 'check-for-changes')
  # %c: Show staged changes in the repository (works with 'check-for-changes')
  #
  # vcs_info_msg_0_ = '%b'
  # vcs_info_msg_1_ = 'x%r' x-prefix prevents creation of a named path
  #                         (AUTO_NAME_DIRS)
  zstyle ':vcs_info:git*' formats '%b' 'x%r'
  # '%a' is for action like 'rebase', 'rebase-i', 'merge'
  zstyle ':vcs_info:git*' actionformats '%b(%a)' 'x%r'

  setopt prompt_subst
  PROMPT='%F{109}%n%f %F{143}%~%f $(prompt_lime_git)$(prompt_lime_symbol) '
}

prompt_lime_setup

{'fraction_non_alphanumeric': 0.15367965367965367, 'fraction_numerical': 0.008658008658008658, 'mean_word_length': 4.267806267806268, 'pattern_counts': {'":': 0, '<': 3, '<?xml version=': 0, '>': 3, 'https://': 0, 'lorem ipsum': 0, 'www.': 0, 'xml': 0}, 'pii_count': 0, 'substrings_counts': 0, 'word_list_counts': {'cursed_substrings.json': 0, 'profanity_word_list.json': 0, 'sexual_word_list.json': 0, 'zh_pornsignals.json': 0}}
{'dir': 'git-commits-cleaned', 'id': '179222', 'n_tokens_mistral': 756, 'n_tokens_neox': 714, 'n_words': 176}